

Combined SG Procedures with ODS to Create Graphs of Individual Data

Howard Liang, Savage, Minnesota

ABSTRACT

The graphical display of individual data is important in understanding the raw data and the relationship between the variables in the data set. You can explore your data to ensure statistical assumptions hold by detecting and excluding outliers if they exist. Since you can visualize what actually happens to individual subjects, you can make your conclusions more convincing in statistical analysis and interpretation of the results. SAS® provides many tools for creating graphs of individual data. In some cases, multiple tools need to be combined to make the specific type of graph you are after. Examples will be used in this paper to show how to create graphs of individual data using the SAS/SG(Statistical Graphics) procedure with ODS(Output Delivery System).

INTRODUCTION

The graphical display of the individual data can be created by producing simple graphs of data such as with a scatter plot and a bar chart. The examples in this paper will explain why sometimes it is not sufficient to create these plots by only using SG procedures and why we need to consider other tools such as ODS. The first example creates a scatter plot with ODS to make annotations work in comparative plot. The second example creates a bar chart with ODS to have multiple plots in a single page.

Example 1

Suppose we want to compare 2 groups of individual data by creating simple scatter plots side by side and drawing a symbol on the outlier.

```
**** CREATE RANDOM SAMPLE, ANNOTATION DATA SETS AND OUTLIER DATA POINTS ****;
%macro sample(seed=, num=, color=%STR());
  data sample&num.;
    do x=1 to 10;
      y=ranuni(&seed.);
      if x=4 then y=y+1;
      group=&num.;
      output;
    end;
  run;
  data anno&num.;
    set sample&num.;
    length label $ 60 y1space x1space $15;
    if x=4 then do;
      retain function 'text' y1space 'datavalue' x1space 'datavalue' width 15;
      y1=y+0.05; x1=x; label="*" ; textsize=20;textcolor="&color.";
      output;
    end;
  run;
%mend sample;
%sample(seed=123, num=1, color=%STR(orange))
%sample(seed=789, num=2, color=%STR(red));

data sample;
  set sample1 sample2;
run;
data anno;
  set anno1 anno2;
run;

**** CREATE PLOTS BY SGPANEL ****;
proc sgpanel data=sample sganno=anno;
  panelby group ;
  scatter x=x y=y;
  rowaxis values=(0 to 2 by 0.5) grid;
  colaxis values=(1 to 10 by 1 ) ;
run;
```

Combined SG Procedures with ODS to Create Graphs of Individual Data, continued

The plot produced by PROC SG PANEL is shown below but the annotated symbol for outlier is not drawn because PROC SG PANEL does not support annotation in data area.




Figure1. Scatter plot using PROC SG PANEL but the annotated symbol for outlier is not drawn

By using PROC SG PLOT and ODS PDF with COLUMNS=2 option, you can have the annotated symbol drawn in the plots. Below is the code that shows how to do this.

```
**** BY USING ODS PDF WITH OPTION COLUMNS=2 ****;
```

```
ods pdf file='c:\temp\test.pdf' columns=2;
proc sgplot data=sample sganno=anno;
  where group=1;
  scatter x=x y=y;
  xaxis values=(1 to 10 by 1);
  yaxis values=(0 to 2 by 0.5) grid;
  title "Group=1";
run;
proc sgplot data=sample sganno=anno;
  where group=2;
  scatter x=x y=y;
  xaxis values=(1 to 10 by 1);
  yaxis values=(0 to 2 by 0.5) grid;
  title "Group=2";
run;
ods pdf close;
```

The plot produced by PROC SG PLOT and ODS is shown below and the symbol for outlier is drawn.




Figure2. Scatter plot using SG PLOT and ODS, annotated symbol is drawn

Combined SG Procedures with ODS to Create Graphs of Individual Data, continued

Example 2

Suppose we want to create simple bar charts to compare 3 different treatments of individual data ordered by response with color of bar by gender and have all 3 plots in one single page.

```
**** CREATE RANDOM SAMPLE ****;
data test;
do subjid=1 to 60;
  if uniform(12345)<=0.33 then Treatment=0;
  else if uniform(12345)<=0.66 then Treatment=1;
  else Treatment=2;
  result=normal(6789);
  Gender=1;
  if mod(subjid,3)=0 then Gender=2;
  output;
end;
run;

proc sort data=test;
by Treatment descending result ;
run;

proc format;
value trt 0="Placebo" 1="Drug A" 2="Drug B";
value sx 1="Female" 2="Male";
run;

proc sgpanel data=test;
panelby treatment/ rows=3;
format treatment trt. gender sx.;
vbar subjid / group=gender response=result categoryorder=respdesc;
colaxis label='Subject' valueattrs=(size=6);
rowaxis label='Response' grid;
run;
```

The plot produced by PROC SG PANEL is shown below but the CATEGORYORDER option is ignored when a group is specified. That column axis has to include all of the values for each of the graphs in each row. As a result, all subject ids in each row will be included in the axis.




Figure3. Bar chart using PROC SG PANEL

Combined SG Procedures with ODS to Create Graphs of Individual Data, continued

By using PROC SGPLOT and ODS PDF with the STARTPAGE=NO option on the ODS PDF statement, you can put all 3 SGPLOT outputs on the same PDF page. Note that in the following code the HEIGHT= and WIDTH= options are used on the ODS Graphics statement to size each graph appropriately.

```
options nodate nonumber;
ods _all_ close;
ods pdf file='c:\temp\test.pdf' notoc dpi=300 startpage=no;

%macro plot(num=, trt=%STR());
  ods graphics / height=3in width=6.5in;
  proc sgplot data=test;
    where treatment=&num.;
    format gender sx.;
    vbar subjid / group=gender response=result ;
    xaxis discreteorder= data label='Subject';
    yaxis label='Response' values=(-2 to 3) grid;
    title "Treatment=&trt.";
  run;
%mend plot;
%plot(num=0, trt=%STR(Placebo))
%plot(num=1, trt=%STR(Drug A))
%plot(num=2, trt=%STR(Drug B));

ods _all_ close;
ods listing;
```

The plot produced by PROC SGPANEL is shown below.

Combined SG Procedures with ODS to Create Graphs of Individual Data, continued




Figure4. Bar chart using SG PLOT and ODS

CONCLUSION

This paper has presented a couple of basic graphs for individual data that are often encountered in clinical trial analysis and reporting. There are many other graphs of data such as histograms, box plots and waterfall charts, and various tools such as GTL (Graph Template Language) in SAS/GRAPH which provide help to anyone who wants to explore data and to achieve more effective communication of clinical trial results. I hope this paper is useful to statistical analysts and others who perform statistical analysis duties in clinical trials or clinical research.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Howard Liang
Address: 12468 Independence Ave
City, State ZIP: Savage, MN 55378
Work Phone: (952) 496-1577
E-mail: hliang5995@yahoo.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.